Quick guide to *Mathematica*

To get help, place the cursor on the function and press F1
To run all commands in one cell, place the cursor there and press Shift + Enter or Enter on the numerical keyboard.
Output of many commands is suppressed by ; at the end of the command. If you want to see the output, delete ;
Argument of functions must be in square brackets [...], braces {...} are for arrays, ranges etc., use double brackets [[...]] to access elements of arrays
expr /. {$x \to a$} means substitute a for x in expr, expr // function means apply function to expr, i.e. it is equivalent to function[expr].
D[f[x],x] = derivative of $f[x]$ with respect to x.
To define a function of x one can use $f[x\_] := 1 + x^2$, notice the underscore and the colon, or # nad & like in $f := 1 + \#^2$ &
N[expr] evaluates expr with machine precision, if you use the decimal point in the expression it will be also evaluated with machine precision
Special characters can be inserted by pressing Esc ... Esc, e.g. Esc p Esc gives π
Clear[...] is used to unset any variables which could have been assigned previously.
Some other useful commands: Simplify, Expand, Factor; Integrate, Series, Sum

Clear all symbols from previous evaluations to avoid problems

```
Clear["Global`*"]
```

Jacobi diagonalization of real symmetric matrix

In[1]:=
```
MyJacobiDiagonalization[A_, niter_] :=
  Module[{EVal, EVec, SumOffDiag, m, θ, t, s, c, τ, i, p, q, r, x, y, nskip},
   m = Length[A];
   EVal = A;
   EVec = IdentityMatrix[m];
   SumOffDiag = ConstantArray[0.0, niter + 1];
   SumOffDiag[[1]] = Total[Abs[UpperTriangularize[EVal, 1]]^2, 2];
   nskip = 0;
   (* only upper triangular part of the matrix is modified *)
   Do[
    Do[
     Do[
      If[Abs[EVal[[p, q]]] < 10.0^(-16),
       nskip++,
       (* θ = cotg (2ϕ) = (c^2 - s^2)/2sc *)
       θ = (EVal[[q, q]] - EVal[[p, p]]) / (2 * EVal[[p, q]]);
       (* t = -θ + Sqrt[θ^2 + 1]; *)
       t = 1 / (Abs[θ] + Sqrt[θ^2 + 1]); If[θ < 0.0, t = -t];
```

```
    c = 1 / Sqrt[1 + t^2]; s = t * c;
    τ = s / (1 + c);

    (* diagonal elements *)
    EVal[[p, p]] = EVal[[p, p]] - t * EVal[[p, q]];
    EVal[[q, q]] = EVal[[q, q]] + t * EVal[[p, q]];
    EVal[[p, q]] = 0.0;
    Do[ (* elements in columns p and q up to the row p-1 *)
     x = EVal[[r, p]]; y = EVal[[r, q]];
     EVal[[r, p]] = x - s * (y + τ * x);
     EVal[[r, q]] = y + s * (x - τ * y),
     {r, 1, p - 1}
    ];
    Do[ (* row p and column q between p and q *)
     x = EVal[[p, r]]; y = EVal[[r, q]];
     EVal[[p, r]] = x - s * (y + τ * x);
     EVal[[r, q]] = y + s * (x - τ * y),
     {r, p + 1, q - 1}
    ];
    Do[ (* rows p and q from q *)
     x = EVal[[p, r]]; y = EVal[[q, r]];
     EVal[[p, r]] = x - s * (y + τ * x);
     EVal[[q, r]] = y + s * (x - τ * y),
     {r, q + 1, m}
    ];
    (* eigenvectors *)
    Do[
     x = EVec[[r, p]]; y = EVec[[r, q]];
     EVec[[r, p]] = x - s * (y + τ * x);
     EVec[[r, q]] = y + s * (x - τ * y),
     {r, 1, m}
    ]
   ],
   {q, p + 1, m}
  ],
  {p, 1, m - 1}
 ];
 SumOffDiag[[i + 1]] = Total[Abs[UpperTriangularize[EVal, 1]]^2, 2],
 {i, 1, niter}
];
Print["Number of skipped elements in Jacobi diagonalization: ", nskip];
Return[{Diagonal[EVal], EVec, SumOffDiag}]
];
```

Application to quantum 1D harmonic oscillator

Exact energies and eigenfunctions of the 1D harmonic oscillator

In[2]:=
```
n = 6;
a = 6;
μ = 1; ω = 1;
V[x_] = 1/2 μ ω^2 x^2;
En[i_] = ω (i + 1/2);
ψ[i_, x_] = 1/Sqrt[2^i i!] (μ ω/π)^(1/4) Exp[-(μ ω x^2)/2] HermiteH[i, x];
Plot[{V[x], Table[En[i], {i, 0, n - 1}], Table[Abs[ψ[i, x]]^2 + En[i], {i, 0, n - 1}]},
 {x, -a, a}, PlotRange → {0, 6}, AxesLabel → {"x", "energy, |ψ|^2"}]
```

Out[8]=



To find eigenenergies and eigenfunctions of the harmonic oscillator numerically we can use any basis satisfying (at least approximately) boundary conditions. If we are interested in a few eigenfunctions with the lowest energies we can safely set boundary conditions

$$\psi(a) = \psi(-a) = 0 \tag{1}$$

for a certain sufficiently large a.

Construction of the Hamiltonian matrix in the sine basis in which all integrals can be evaluated in the closed form:

In[9]:=
```
n = 10;
Clear[H];
ϕ[i_, x_] = Sin[2 π (x + a) i / (4 a)] / Sqrt[a];
KE[i_, j_] = If[i == j, 1/(2 μ) (i Pi/(2 a))^2, 0];
PE[i_, j_] =
  If[i == j, -a^2 (1/(i^2 π^2) - 1/6), If[Mod[i + j, 2] == 1, 0, 4 a^2/π^2 (1/(i - j)^2 - 1/(i + j)^2)]];
H = ConstantArray[0, {n, n}];
Do[
  Do[
   H[[i, j]] = KE[i, j] + PE[i, j]
   (* Integrate[-ϕ[i,x]D[ϕ[j,x],{x,2}]/(2m)(* +ϕ[i,x]V[x]ϕ[j,x]*),{x,-a,a}]*),
   {j, 1, n}
  ],
  {i, 1, n}
 ];
N[H] // MatrixForm
```

Out[16]//MatrixForm=

2.3867068486	0.	2.73567195834	0.	0.506605918212	0.
0.	5.22518718612	0.	3.24227787655	0.	0.6839179895
2.73567195834	0.	5.90314040296	0.	3.41958994793	0.
0.	3.24227787655	0.	6.32033869242	0.	3.501660106
0.506605918212	0.	3.41958994793	0.	6.71083398871	0.
0.	0.683917989586	0.	3.50166010668	0.	7.132379366
0.177312071374	0.	0.765988148336	0.	3.54624142748	0.
0.	0.259382230124	0.	0.810569469139	0.	3.573122557
0.0820701587503	0.	0.303963550927	0.	0.837450599493	0.
0.	0.126651479553	0.	0.330844681281	0.	0.8548974869

Diagonalization using the Jacobi method:

In[17]:=
```
{Energies, Coefficients, Sums} = MyJacobiDiagonalization[N[H], 4];
Print["Energies:"]
Energies
Print["Coefficients:"]
Coefficients // MatrixForm
Print["Sums of off-diagonal elements after each sweep:"]
Sums
```

```
Number of skipped elements in Jacobi diagonalization: 100
```

```
Energies:
```

Out[19]=

```
{0.50104958805, 1.50410432891, 14.0574364841, 14.8092333568, 2.58284114737,
 3.62555936997, 5.22687945099, 6.31244437983, 8.96803264363, 9.95328641273}
```

```
Coefficients:
```

Out[21]//MatrixForm=

```
( 0.743682762039        0.        0.111346166592        0.       -0.473403451431        0.
        0.        0.498690486113        0.        0.177275037187        0.       -0.531557
 -0.564989219679        0.        0.326413236198        0.       -0.114860190659        0.
        0.       -0.660227011262        0.        0.351239460359        0.        0.204710
  0.325529735666        0.        0.508923052788        0.        0.596021322138        0.
        0.        0.496603599552        0.        0.504563875225        0.        0.446609
 -0.141381501882        0.        0.603832964509        0.       -0.578913890865        0.
        0.       -0.249797178391        0.        0.587319592465        0.       -0.612981
 0.0420094419831        0.        0.507384969952        0.        0.268911880169        0.
        0.        0.0799645373903        0.        0.495656529062        0.        0.316756
```

```
Sums of off-diagonal elements after each sweep:
```

Out[23]=

$$\left\{96.8886114064,\ 7.66678801731,\ 0.11407855087,\ 0.0000259882514481,\ 3.09543536242\times 10^{-21}\right\}$$

In[24]:=

```
index = Ordering[Energies];
m = 8;
Plot[{V[x], Table[En[i], {i, 0, m - 1}],
  Table[Abs[ψ[i, x]]^2 + En[i], {i, 0, m - 1}], Table[Energies⟦index⟦i⟧⟧, {i, 1, m}],
  Table[Abs[Sum[Coefficients⟦j, index⟦i⟧⟧ ϕ[j, x], {j, 1, n}]]^2 + Energies⟦index⟦i⟧⟧,
   {i, 1, m}]}, {x, -a, a}, PlotRange → {0, m + 3}, AxesLabel → {"x", "energy, |ψ|^2"}]
```

Out[26]=